

Tcl Variable Utilities

Version 0.3

Alex Baker

<https://github.com/ambaker1/vutil>

June 7, 2023

Abstract

This package provides various utilities for working with variables in Tcl, including read-only variables, TclOO garbage collection, and an object-variable type system.

Printing Variables to Screen

The *pvar* command is a short-hand function for printing the name and values of Tcl variables.

```
pvar $name1 $name2 ...
```

`$name1 $name2 ...` Name(s) of variables to print

Example 1: Printing variables to screen

Code:

```
set a 5
set b 7
set c(1) 5
set c(2) 6
pvar a b c
```

Output:

```
a = 5
b = 7
c(1) = 5
c(2) = 6
```

Initializing Local Namespace Variables

The command *local* is the counterpart to the Tcl *global* command, and creates local variables linked to variables in the current namespace, by simply calling the Tcl *variable* command multiple times.

```
local $name1 $name2 ...
```

`$name1 $name2 ...` Name(s) of variables to initialize

Example 2: Access namespace variables in a procedure

Code:

```
# Define global variables
global a b c
set a 1
set b 2
set c 3
namespace eval ::foo {
    # Define local variables
    local a b c
    set a 4
    set b 5
    set c 6
}
proc ::foo::bar1 {} {
    # Access global variables
    global a b c
    list $a $b $c
}
proc ::foo::bar2 {} {
    # Access local variables
    local a b c
    list $a $b $c
}
puts [::foo::bar1]; # global a b c
puts [::foo::bar2]; # local a b c
```

Output:

```
1 2 3
4 5 6
```

Default Values

The command *default* assigns values to variables if they do not exist.

```
default $varName $value
```

\$varName Name of variable to set

\$value Default value for variable

The example below shows how default values are only applied if the variable does not exist.

Example 3: Variable defaults

Code:

```
set a 5
default a 7
puts $a
unset a
default a 7
puts $a
```

Output:

```
5
7
```

Variable Locks

The command *lock* uses Tcl variable traces to make a read-only variable.

```
lock $varName <$value>
```

\$varName Variable name to lock.

\$value Value to lock variable at. Default self-locks (uses current value).

The command *unlock* unlocks previously locked variables so that they can be modified again.

```
unlock $name1 $name2 ...
```

\$name1 \$name2 ... Variables to unlock.

Example 4: Variable locks

Code:

```
lock a 5
set a 7
puts $a
unlock a
set a 7
puts $a
```

Output:

```
5
7
```

Variable-Object Ties

As of Tcl version 8.6, there is no garbage collection for Tcl objects, they have to be removed manually with the “destroy” method. The command *tie* is a solution for this problem, using variable traces to destroy the corresponding object when the variable is unset or modified. Tie is separate from lock; a tie will override a lock, and a lock will override a tie.

```
tie $varName <$object>
```

\$varName Variable name to tie to object.

\$object Object to tie variable to. Default self-ties (uses current value).

In similar fashion to *unlock*, tied variables can be untied with the command *untie*.

```
untie $name1 $name2 ...
```

\$name1 \$name2 ... Variables to untie.

Example 5: Variable-object ties

Code:

```
oo::class create foo {
    method hi {} {
        puts hi
    }
}
tie a [foo create bar]
set b $a; # alias variable
unset a; # triggers ``destroy''
$b hi; # throws error
```

Output:

```
invalid command name "::bar"
```

Variable-Object Links

The command *link* links a global variable to a TclOO object, using the name of the object as the variable name. The value of the linked object variable is accessed by calling the TclOO object with no arguments (the “unknown” method), and writing to the object-variable calls the “=” method. Unsetting the linked object variable also calls the object’s “destroy” method, and destroying the object unsets the linked object variable. Linked object-variables are unlinked when the object is destroyed, but can also be unlinked with the command *unlink*.

```
link $Object
```

```
unlink $Object ...
```

\$Object ... Object(s) to link/unlink.

Example 6: Linking object variable

Code:

```
::oo::class create number {
    variable value
    constructor {args} {
        my = {*} $args
    }
    method unknown {args} {
        if {[llength $args] == 0} {
            return $value
        }
        next {*} $args
    }
    unexport unknown
    method = {args} {
        set value [uplevel 1 expr $args]
    }
    export =
}
link [tie a [number new 5]]; # garbage collection and obj-var link
puts [$a]; # 5
$a = 10 * [$a]
puts [$a]; # 50
incr $a
puts [subst $$a]; # 51
```

Output:

```
5
50
51
```

Object Variable Class

The `var TclOO` class creates linked object variables that are tied to local variables, using *link* and *tie*.

```
var new $refName <<"="> $value | "<-" $varObj>
var create $objName $refName <<"="> $value | "<-" $varObj>
```

<code>\$objName</code>	Explicit name for object.
<code>\$refName</code>	Name of reference variable to tie to object.
<code>\$value</code>	Value to set object variable to (using the optional “=” keyword).
<code>\$varObj</code>	Other object variable to copy value from (using the “<-” keyword).

Example 7: Object variables with garbage collection

Code:

```
# Example showing how object variables behave in procedures
proc foo {value} {
    obj create myObj a = $value
    puts $a; # ::myObj
    puts [info object isa object $a]; # 1
    append $a { world}
    puts [$a]; # hello world
    return $a; # Returns name of object, not value
}
set a [foo hello]; # Not the same "a"
puts $a; # ::myObj
puts [info object isa object $a]; # 0 (object was deleted when procedure returned)
```

Output:

```
::myObj
1
hello world
::myObj
0
```


Object Variable Methods

If the variable object is called with no method (e.g. [`$varObj`]), it will simply return the value of the object variable. In addition, object variables have the following methods:

The method “=” assigns the value of the object variable.

```
$varObj = $value
```

`$value` Value to set object variable to.

The method “<-” assigns the value of the object directly from another object variable of the same class.

```
$varObj <- $otherVarObj
```

`$otherVarObj` Other object variable to assign value from.

The method “-->” copies the object variable.

```
$varObj --> $refName <$arg ...>
```

`$refName` Name of reference variable to tie to object.

`$arg ...` Additional arguments to pass to Tcl *oo::copy* command.

The method *info* accesses object variable metadata. Fields “exists” and “type” always exist, and “value” is populated when the variable is initialized.

```
$varObj info <$field>
```

`$field` Info field to get. Default returns dictionary of all info.

Example 8: Copying variables

Code:

```
var new x
$x --> y
$x = 5; # does not affect copy
puts [$x info]
puts [$y info]
```

Output:

```
exists 1 type {} value 5
exists 0 type {}
```

Object Variable Types

The *var* TclOO class acts as a superclass for a pure-Tcl type system, managed through the metaclass *type*.

```
type $method $arg ...
```

<code>\$method</code>	Method name.
<code>\$arg ...</code>	Arguments for method.

New type classes can be created using the method *new*. The resulting class, named “`::vutil::type.$type`”, uses “`::vutil::var`” as a superclass.

```
type new $type $arg ...
```

<code>\$type</code>	Name of type.
<code>\$arg ...</code>	Class definition arguments.

Existing classes can be added to the type library directly with the method *add*. These classes do not need to be subclasses of `::vutil::var`, but it is recommended to utilize `::vutil::var` as a superclass.

```
type add $type $class
```

<code>\$type</code>	Name of type.
<code>\$class</code>	Type class.

A list of all defined types can be queried with the method *names*.

```
type names
```

Types can be removed with the method *remove*, the existence of a type can be queried with the method *exists*, and the class associated with a type can be queried with the method *class*.

```
type remove $type
```

```
type exists $type
```

```
type class $type
```

<code>\$type</code>	Name of type.
---------------------	---------------

Creating Type Variables

Then, using the types defined by *type*, the command *new* creates a new variable object of specified type.

```
new $type $refName <<"="> $value | "<-" $varObj>
```

\$type	Name of type.
\$refName	Name of reference variable to tie to object.
\$value	Value to set object variable to (using the optional “=” keyword).
\$varObj	Other object variable to copy value from (using the “<-” keyword).

Example 9: New type and object variables

Code:

```
type new str {
  method info {args} {
    set (length) [my length]
    next {*} $args
  }
  method print {} {
    puts $(value)
  }
  method length {} {
    string length $(value)
  }
  method @ {i} {
    string index $(value) $i
  }
  export @
}
new str x
set $x {hello world}
puts [$x length]
puts [$x info]
puts [$x @ end]
$x print
```

Output:

```
11
exists 1 length 11 type str value {hello world}
d
hello world
```

Type Library

This package provides a few basic object variable types: *var*, *string*, *bool*, *int*, *float*, *list*, and *dict*. If you want to create your own datatype, check out the source code implementation for API documentation.

Type “var”

The type *var* is just an alternative way to create an object variable (same syntax as “new” method for *var* class). It does not have any additional metadata or methods.

```
new var $refName <<“=”> $value | “<-” $varObj>
```

\$refName	Name of reference variable to tie to object.
\$value	Value to set object variable to (using the optional “=” keyword).
\$varObj	Other “var” to copy value from (using the “<-” keyword).

Type “string”

The type *string* does not do any validation on input (because in Tcl, “everything is a string”), but additionally provides methods for getting string length and string index, and adds the field “length” to the variable info.

```
new string $refName <<“=”> $value | “<-” $stringObj>
```

\$refName	Name of reference variable to tie to object.
\$value	Value to set object variable to (using the optional “=” keyword).
\$stringObj	Other “string” to copy value from (using the “<-” keyword).

The method *length* returns the string length. This is the same as `$stringObj info length`.

```
$stringObj length
```

The method “@” returns the character at the specified index.

```
$stringObj @ $i
```

\$i	String index.
------------	---------------

Type “bool” (boolean)

The type *bool* passes input through the Tcl *expr* command and verifies that the result is indeed boolean.

```
new bool $refName << "=" > $expr | "<-" $boolObj>
```

\$refName	Name of reference variable to tie to object.
\$expr	Tcl math expression to evaluate (using the optional “=” keyword).
\$boolObj	Other “bool” to copy value from (using the “<-” keyword).

In addition to the standard object variable methods, the “bool” type provides a shorthand if-statement control flow method:

```
$boolObj "?" $body1 <":" $body2>
```

\$body1	Body to evaluate if boolean is true.
\$body2	Body to evaluate if boolean is false (optional, required with “:” keyword).

Example 10: String and boolean example

Code:

```
# proc with types
proc foo {a b c} {
    new string a $a
    new string b $b
    new bool c $c
    $c ? {$a} : {$b}
}
puts [foo hello world true]; # hello
puts [foo hello world false]; # world
```

Output:

```
hello
world
```

Type “int” (integer)

The type *int* passes input through the Tcl *expr* command and verifies that the result is indeed an integer.

```
new int $refName <<“=”> $expr | “<-” $intObj>
```

\$refName	Name of reference variable to tie to object.
\$expr	Tcl math expression to evaluate (using the optional “=” keyword).
\$intObj	Other “int” to copy value from (using the “<-” keyword).

In addition to the standard assignment operator “=”, the “int” type provides the increment/decrement assignment operators “+=” and “-=”.

```
$intObj $op $expr
```

\$op	Assignment operator: “+=” Increments the value of \$expr to the variable. “-=” Decrements the value of \$expr to the variable.
\$expr	Tcl math expression to evaluate.

There are also two short-hand increment/decrement operators, “++” and “--”, which simply increment or decrement the integer object by 1.

```
$intObj ++
```

```
$intObj --
```

Example 11: Integer example (for loop)

Code:

```
for {new int i 0} {[ $i] < 3} { $i ++} {  
    puts [ $i]  
}
```

Output:

```
0  
1  
2
```

Type “float” (double-precision floating-point decimal)

The type *float* passes input through the Tcl *expr* command and the `::tcl::mathfunc::double` command, ensuring that the value is a double-precision floating-point number.

```
new float $refName <<“=”> $expr | “<-” $floatObj>
```

\$refName	Name of reference variable to tie to object.
\$expr	Tcl math expression to evaluate (using the optional “=” keyword).
\$floatObj	Other “float” to copy value from (using the “<-” keyword).

In addition to the standard assignment operator “=”, the “float” type provides the assignment operators “+=”, “-=”, “*=”, and “/=”.

```
$floatObj $op $expr
```

\$op	Assignment operator: “+=” Adds the value of \$expr to the variable. “-=” Subtracts the value of \$expr to the variable. “*=” Multiplies the value of \$expr to the variable. “/=” Divides the variable by the value of \$expr .
\$expr	Tcl math expression to evaluate.

Example 12: Float example (procedure with type assertion)

Code:

```
# Harmonic mean of two numbers (converts to float)
proc hmean {x y} {
    new float x $x
    new float y $y
    if {[$x] == 0 || [$y] == 0} {
        return 0
    }
    expr {2*[$x]*[$y]/([$x] + [$y])}
}
puts [hmean 1 2]; # 1.3333
```

Output:

```
1.3333333333333333
```

Type “list”

The type *list* validates that the input is a list, and provides methods for list indexing/setting and getting list length. Additionally, it adds the field “length” to the variable info.

```
new list $refName << "=" > $value | "<-" $listObj>
```

\$refName	Name of reference variable to tie to object.
\$value	Value to set object variable to (using the optional “=” keyword).
\$listObj	Other “list” to copy value from (using the “<-” keyword).

The method *\$listObj length* returns the list length. This is the same as *\$listObj info length*.

```
$listObj length
```

The method *\$listObj @* acts as both list indexing and list setting, depending on whether the “=” keyword is used. If indexing, it returns the value at the specified index. If setting, it sets the value and returns the object name.

```
$listObj @ $i ... < "=" $value>
```

\$i ...	List indices.
\$value	Value to set.

Example 13: List example

Code:

```
new list list1 = {hello world}
puts [$list1 length]; # 2
$list1 @ 0 = "hey"
$list1 @ 1 = "there"
$list1 @ end+1 = "world"
puts [$list1 @ end]; # world
puts [$list1 info]; # exists 1 length 3 type list value {hey there world}
```

Output:

```
2
world
exists 1 length 3 type list value {hey there world}
```


Type “dict”

The type *dict* validates that the input is a Tcl dictionary, and provides methods for getting/setting/unsetting dictionary values, checking if values exist, and getting the dictionary size. Additionally, it adds the field “size” to the variable info.

```
new dict $refName <<“=”> $value | “<-” $tObj>
```

\$refName	Name of reference variable to tie to object.
\$value	Value to set object variable to (using the optional “=” keyword).
\$dictObj	Other “dict” to copy value from (using the “<-” keyword).

The method *\$dictObj size* returns the dictionary size. This is the same as *\$dictObj info size*.

```
$dictObj size
```

The methods *set* and *unset* set and unset values in the dictionary, and return the object name. The method *get* returns values in the dictionary, and the method *exists* returns whether the key pairing exists.

```
$dictObj set $key ... $value
```

```
$dictObj unset $key ...
```

```
$dictObj get $key ...
```

```
$dictObj exists $key ...
```

\$key ...	Dictionary keys.
\$value	Value to set.

Example 14: Dictionary example

Code:

```
# Create dictionary record
new dict record = {
    name {John Doe}
    address {
        streetAddress {123 Main Street}
        city {New York}
        state {NY}
        zip {10001}
    }
    phone {555-1234}
}

# Get values
puts [$record size]; # Number of keys (3)
puts [$record get name]; # John Doe
# Set/unset and get
$record set address street [$record get address streetAddress]
$record unset address streetAddress
puts [$record get address street]; # 123 Main Street
puts [$record exists address streetAddress]; # 0
```

Output:

```
3
John Doe
123 Main Street
0
```